
Lightweight tracing, debugging and profiling tool, which collects traces in an ETS table, putting minimal impact on your system.
After collecting the traces, you can query and analyse them.
By separating data collection from analysis, this tool helps you limit unnecessary repetition and guesswork.
There is ExDoctor for Elixir as well.
Quick start
To quickly try it out right now, copy & paste the following to your Erlang shell:
P = "/tmp/tr.erl", ssl:start(), inets:start(), {ok, {{_, 200, _}, _, Src}} = httpc:request("https://git.io/fj024"), file:write_file(P, Src), {ok, tr, B} = compile:file(P, binary), code:load_binary(tr, P, B), rr(P), tr:start().
This snippet downloads, compiles and starts the tr module from the master branch.
Your Erlang Doctor is now ready to use!
The easiest way to use it is the following:
tr:trace([your_module]).
your_module:some_function().
tr:select().
You should see the collected traces for the call and return of your_module:some_function/0.
This compact tool is capable of much more - see below.
Include it as a dependency
To avoid copy-pasting the snippet shown above, you can include erlang_doctor in your dependencies in rebar.config.
There is a Hex package as well.
Use it during development
You can make Erlang Doctor available in the Erlang/Rebar3 shell during development by cloning it to ERLANG_DOCTOR_PATH,
calling rebar3 compile, and loading it in your ~/.erlang file:
code:add_path("ERLANG_DOCTOR_PATH/erlang_doctor/_build/default/lib/erlang_doctor/ebin").
code:load_file(tr).
Tracing: data collection
The test suite helpers from tr_SUITE.erl are used here as examples.
You can follow these examples on your own - just call rebar3 as test shell in ERLANG_DOCTOR_PATH.
Setting up: start, start_link
The first thing to do is to start the tracer with tr:start/0.
There is also tr:start/1, which accepts a map of options, including:
	tab: collected traces are stored in an ETS table with this name (default: trace),
	limit: maximum number of traces in the table - when it is reached, tracing is stopped (default: no limit).

There are tr:start_link/0 and tr:start_link/1 as well, and they are intended for use with the whole erlang_doctor application.
For this tutorial we start the tr module in the simplest way:
1> tr:start().
{ok, <0.218.0>}
Tracing with trace
To trace function calls for given modules, use tr:trace/1, providing a list of traced modules:
2> tr:trace([tr_SUITE]).
ok
You can provide {Module, Function, Arity} tuples in the list as well.
The function tr:trace_app/1 traces an application, and tr:trace_apps/1 traces multiple ones.
If you need to trace an application and some additional modules, use tr:app_modules/1 to get the list of modules for an application:
tr:trace([Module1, Module2 | tr:app_modules(YourApp)]).
If you want to trace selected processes instead of all of them, you can use tr:trace/2:
tr:trace([Module1, Module2], [Pid1, Pid2]).
The tr:trace/1 function accepts a map of options, which include:
	modules: a list of module names or {Module, Function, Arity} tuples. The list is empty by default.
	pids: a list of Pids of processes to trace, or the atom all (default) to trace all processes.
	msg: none (default), all, send or recv. Specifies which message events will be traced. By default no messages are traced.
	msg_trigger: after_traced_call (default) or always. By default, traced messages in each process are stored after the first traced function call in that process. The goal is to limit the number of traced messages, which can be huge in the entire Erlang system. If you want all messages, set it to always.

Calling the traced function
Now we can call some functions - let's trace the following function call.
It calculates the factorial recursively and sleeps 1 ms between each step.
3> tr_SUITE:sleepy_factorial(3).
6
Stopping tracing
You can stop tracing with the following function:
4> tr:stop_tracing().
ok
It's good to stop it as soon as possible to avoid accumulating too many traces in the ETS table.
Usage of tr on production systems is risky, but if you have to do it, start and stop the tracer in the same command,
e.g. for one second with:
tr:trace(Modules), timer:sleep(1000), tr:stop_tracing().
Debugging: data analysis
The collected traces are stored in an ETS table (default name: trace).
They are stored as tr records with the following fields:
	index: trace identifier, auto-incremented for each received trace.
	pid: process identifier associated with the trace.
	event: call, return or exception for function traces; send or recv for messages.
	mfa: {Module, Function, Arity} for function traces; no_mfa for messages.
	data: argument list (for calls), returned value (for returns) or class and value (for exceptions).
	timestamp in microseconds.
	info: For function traces and recv events it is no_info. For send events it is a {To, Exists} tuple, where To is the recipient pid, and Exists is a boolean indicating if the recipient process existed.

It's useful to read the record definitions before trace analysis:
5> rr(tr).
[node,tr]
The snippet shown at the top of this page includes this already.
Trace selection: select
Use tr:select/0 to select all collected traces.
6> tr:select().
[#tr{index = 1,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [3],
     ts = 1705475521743239,info = no_info},
 #tr{index = 2,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [2],
     ts = 1705475521744690,info = no_info},
 #tr{index = 3,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [1],
     ts = 1705475521746470,info = no_info},
 #tr{index = 4,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [0],
     ts = 1705475521748499,info = no_info},
 #tr{index = 5,pid = <0.395.0>,event = return,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = 1,ts = 1705475521750451,info = no_info},
 #tr{index = 6,pid = <0.395.0>,event = return,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = 1,ts = 1705475521750453,info = no_info},
 #tr{index = 7,pid = <0.395.0>,event = return,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = 2,ts = 1705475521750454,info = no_info},
 #tr{index = 8,pid = <0.395.0>,event = return,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = 6,ts = 1705475521750455,info = no_info}]
The tr:select/1 function accepts a fun that is passed to ets:fun2ms/1.
This way you can limit the selection to specific items and select only some fields from the tr record:
7> tr:select(fun(#tr{event = call, data = [N]}) -> N end).
[3, 2, 1, 0]
Use tr:select/2 to further filter the results by searching for a term in #tr.data (recursively searching in lists, tuples and maps).
8> tr:select(fun(T) -> T end, 2).
[#tr{index = 2,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [2],
     ts = 1705475521744690,info = no_info},
 #tr{index = 7,pid = <0.395.0>,event = return,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = 2,ts = 1705475521750454,info = no_info}]
Trace filtering: filter
Sometimes it might be easier to use tr:filter/1, because it can accept any function as the argument.
You can use tr:contains_data/2 to search for the same term as in the example above.
9> Traces = tr:filter(fun(T) -> tr:contains_data(2, T) end).
[#tr{index = 2,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [2],
     ts = 1705475521744690,info = no_info},
 #tr{index = 7,pid = <0.395.0>,event = return,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = 2,ts = 1705475521750454,info = no_info}]
The provided function is a predicate, which has to return true for the matching traces.
For other traces it can return another value, or even raise an exception:
10> tr:filter(fun(#tr{data = [2]}) -> true end).
[#tr{index = 2,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [2],
     ts = 1705475521744690,info = no_info}]
There is also tr:filter/2, which can be used to search in a different table than the current one - or in a list:
11> tr:filter(fun(#tr{event = call}) -> true end, Traces).
[#tr{index = 2,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [2],
     ts = 1705475521744690,info = no_info}]
There are also additional ready-to-use predicates besides tr:contains_data/2:
	tr:match_data/2 performs a recursive check for terms like tr:contains_data/2,
but instead of checking for equality, it applies the predicate function
provided as the first argument to each term, and returns true if the predicate returned true
for any of them. The predicate can return any other value or fail for non-matching terms.
	tr:contains_val/2 is like tr:contains_data/2, but the second argument is the Data term itself rather than a trace record with data.
	tr:match_val/2 is like tr:match_data/2, but the second argument is the Data term itself rather than a trace record with data.

By combining these predicates, you can search for complex terms, e.g.
the following expression returns trace records that contain any (possibly nested in tuples/lists/maps)
3-element tuples with a map as the third element - and that map has to contain the atom error
(possibly nested in tuples/lists/maps).
tr:filter(fun(T) -> tr:match_data(fun({_, _, Map = #{}}) -> tr:contains_val(error, Map) end, T) end).
Tracebacks for filtered traces: tracebacks
To find the tracebacks (stack traces) for matching traces, use tr:tracebacks/1:
12> tr:tracebacks(fun(#tr{data = 1}) -> true end).
[[#tr{index = 3,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [1],
      ts = 1705475521746470,info = no_info},
  #tr{index = 2,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [2],
      ts = 1705475521744690,info = no_info},
  #tr{index = 1,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [3],
      ts = 1705475521743239,info = no_info}]]
Note, that by specifying data = 1 we are only matching return traces, as call traces always have a list in data.
Only one traceback is returned. It starts with a call that returned 1. What follows is the stack trace for this call.
One can notice that the call for 0 also returned 1, but the call tree got pruned - whenever two tracebacks overlap, only the shorter one is left.
You can change this by returning tracebacks for all matching traces even if they overlap, setting the output option to all. All options are specified in the second argument, which is a map:
13> tr:tracebacks(fun(#tr{data = 1}) -> true end, #{output => all}).
[[#tr{index = 4,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [0],
      ts = 1705475521748499,info = no_info},
  #tr{index = 3,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [1],
      ts = 1705475521746470,info = no_info},
  #tr{index = 2,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [2],
      ts = 1705475521744690,info = no_info},
  #tr{index = 1,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [3],
      ts = 1705475521743239,info = no_info}],
 [#tr{index = 3,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [1],
      ts = 1705475521746470,info = no_info},
  #tr{index = 2,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [2],
      ts = 1705475521744690,info = no_info},
  #tr{index = 1,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [3],
      ts = 1705475521743239,info = no_info}]]
The third possibility is output => longest which does the opposite of pruning, leaving only the longest tracabecks when they overlap:
14> tr:tracebacks(fun(#tr{data = 1}) -> true end, #{output => longest}).
[[#tr{index = 4,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [0],
      ts = 1705475521748499,info = no_info},
  #tr{index = 3,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [1],
      ts = 1705475521746470,info = no_info},
  #tr{index = 2,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [2],
      ts = 1705475521744690,info = no_info},
  #tr{index = 1,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [3],
      ts = 1705475521743239,info = no_info}]]
Possible options for tr:tracebacks/2 include:
	tab is the table or list which is like the second argument of tr:filter/2,
	output - shortest (default), all, longest - see above.
	format - list (default), tree - returns a list of (possibly merged) call trees instead of tracebacks, root - returns a list of root calls. Trees and roots don't distinguish between all and longest output formats. Using root is equivalent to using tree, and then calling tr:roots/1 on the results. There is also tr:root/1 for a single tree.
	order - top_down (default), bottom_up - call order in each tracaback; only for the list format.
	limit - positive integer or infinity (default) - limits the number of matched traces. The actual number of tracebacks returned can be smaller unless output => all

There are also functions tr:traceback/1 and tr:traceback/2. They set limit to one and return only one trace if it exists. The options for tr:traceback/2 are the same as for tr:traceback/2 except limit and format (which are not supported). Additionally, it is possible to pass a tr record (or an index) as the first argument to tr:traceback/1 or tr:traceback/2 to obtain the traceback for the provided trace event.
Trace ranges for filtered traces: ranges
To get a list of traces between each matching call and the corresponding return, use tr:ranges/1:
15> tr:ranges(fun(#tr{data = [1]}) -> true end).
[[#tr{index = 3,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [1],
      ts = 1705475521746470,info = no_info},
  #tr{index = 4,pid = <0.395.0>,event = call,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = [0],
      ts = 1705475521748499,info = no_info},
  #tr{index = 5,pid = <0.395.0>,event = return,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = 1,ts = 1705475521750451,info = no_info},
  #tr{index = 6,pid = <0.395.0>,event = return,
      mfa = {tr_SUITE,sleepy_factorial,1},
      data = 1,ts = 1705475521750453,info = no_info}]]
There is also tr:ranges/2 - it accepts a map of options, including:
	tab is the table or list which is like the second argument of tr:filter/2,
	max_depth is the maximum depth of nested calls. A message event also adds 1 to the depth.
  You can use #{max_depth => 1} to see only the top-level call and the corresponding return.
	output - all (default), complete or incomplete - decides whether the output should contain
  complete and/or incomplete ranges. A range is complete if the root call has a return.
  For example, you can use #{output => incomplete} to see only the traces with missing returns.

When you combine the options into #{output => incomplete, max_depth => 1},
you get all the calls which didn't return (they were still executing when tracing was stopped).
There are two additional functions: tr:range/1 and tr:range/2, which return only one range if it exists. It is possible to pass a tr record or an index as the first argument to tr:range/1 or tr:range/2 as well.
Calling a function from a trace: do
It is easy to replay a particular function call with tr:do/1:
16> [T] = tr:filter(fun(#tr{data = [3]}) -> true end).
[#tr{index = 1,pid = <0.395.0>,event = call,
     mfa = {tr_SUITE,sleepy_factorial,1},
     data = [3],
     ts = 1705475521743239,info = no_info}]
17> tr:do(T).
6
This is useful e.g. for checking if a bug has been fixed without running the whole test suite.
This function can be called with an index as the argument.
Browsing traces: lookup, prev, next
Use tr:lookup/1 to obtain the trace record for an index. Also, given a trace record or an index,
you can obtain the next trace record with tr:next/1, or the previous one with tr:prev/1.
18> T2 = tr:next(T).
#tr{index = 2,pid = <0.424.0>,event = call,
    mfa = {tr_SUITE,sleepy_factorial,1},
    data = [2],
    ts = 1752566205078116,info = no_info}
19> T = tr:prev(T2).
#tr{index = 1,pid = <0.424.0>,event = call,
    mfa = {tr_SUITE,sleepy_factorial,1},
    data = [3],
    ts = 1752566205069393,info = no_info}
When there is no trace to return, not_found error is raised:
20> tr:prev(T).
** exception error: not_found
     in function  tr:prev/3
        called as tr:prev(1,#Fun<tr.15.31036569>,trace)
There are also more advanced variants of these fucntions: tr:next/2 and tr:prev/2.
As their second argument, they take a map of options, including:
	tab is the table or list (like the second argument of tr:filter/2),
	pred is a predicate function that should return true for a matching trace record.
  For other arguments, it can return a different value or fail.
  When used, tab will be traversed until a matching trace is found.

There are also functions tr:seq_next/1 and tr:seq_prev/1.
Given a trace record or an index, they first check the Pid of the process for that trace record,
and then they return next/previous trace record with the same Pid.
This effect could be achieved with a predicate function: fun(#tr{pid = P}) -> P =:= Pid end,
but these utility functions are much more handy.
Profiling
You can quickly get a hint about possible bottlenecks and redundancies in your system with function call statistics.
Call statistics: call_stat
The argument of tr:call_stat/1 is a function that returns a key by which the traces are grouped.
The simplest way to use this function is to look at the total number of calls and their time.
To do this, we group all calls under one key, e.g. total:
21> tr:call_stat(fun(_) -> total end).
#{total => {4,7216,7216}}
Values of the returned map have the following format (time is in microseconds):
{call_count(), acc_time(), own_time()}
In the example there are four calls, which took 7216 microseconds in total.
For nested calls we only take into account the outermost call, so this means that the whole calculation took 7.216 ms.
Let's see how this looks like for individual steps - we can group the stats by the function argument:
22> tr:call_stat(fun(#tr{data = [N]}) -> N end).
#{0 => {1,1952,1952},
  1 => {1,3983,2031},
  2 => {1,5764,1781},
  3 => {1,7216,1452}}
You can use the provided function to do filtering as well:
23> tr:call_stat(fun(#tr{data = [N]}) when N < 3 -> N end).
#{0 => {1,1952,1952},1 => {1,3983,2031},2 => {1,5764,1781}}
Sorted call statistics: sorted_call_stat
You can sort the call stat by accumulated time (descending) with tr:sorted_call_stat/1:
24> tr:sorted_call_stat(fun(#tr{data = [N]}) -> N end).
[{3,1,7216,1452},
 {2,1,5764,1781},
 {1,1,3983,2031},
 {0,1,1952,1952}]
The first element of each tuple is the key, the rest is the same as above.
To pretty-print it, use tr:print_sorted_call_stat/2.
The second argument limits the table row number, e.g. we can only print the top 3 items:
25> tr:print_sorted_call_stat(fun(#tr{data = [N]}) -> N end, 3).
3  1  7216  1452
2  1  5764  1781
1  1  3983  2031
ok
Call tree statistics: top_call_trees
The function tr:top_call_trees/0 makes it possible to detect complete call trees that repeat several times,
where corresponding function calls and returns have the same arguments and return values, respectively.
When such functions take a lot of time and do not have useful side effects, they can be often optimized.
As an example, let's trace the call to a function which calculates the 4th element of the Fibonacci Sequence
in a recursive way. The trace table should be empty, so let's clean it up first:
26> tr:clean().
ok
27> tr:trace([tr_SUITE]).
ok
28> tr_SUITE:fib(4).
3
29> tr:stop_tracing().
ok
Now it is possible to print the most time consuming call trees that repeat at least twice:
30> tr:top_call_trees().
[{13,2,
  #node{module = tr_SUITE,function = fib,
        args = [2],
        children = [#node{module = tr_SUITE,function = fib,
                          args = [1],
                          children = [],
                          result = {return,1}},
                    #node{module = tr_SUITE,function = fib,
                          args = [0],
                          children = [],
                          result = {return,0}}],
        result = {return,1}}},
 {5,3,
  #node{module = tr_SUITE,function = fib,
        args = [1],
        children = [],
        result = {return,1}}}]
The resulting list contains tuples {Time, Count, Tree} where Time is the accumulated time (in microseconds) spent in the tree,
and Count is the number of times the tree repeated. The list is sorted by Time, descending.
In the example above fib(2) was called twice and fib(1) was called 3 times,
what already shows that the recursive implementation is suboptimal.
There is also tr:top_call_trees/1, which takes a map of options, including:
	output is reduced by default, but it can be set to complete where subtrees of already listed trees are also listed.
	min_count is the minimum number of times a tree has to occur to be listed, the default is 2.
	min_time is the minimum accumulated time for a tree, by default there is no minimum.
	max_size is the maximum number of trees presented, the default is 10.

As an exercise, try calling tr:top_call_trees(#{min_count => 1000}) for fib(20).
Exporting and importing traces
To get the current table name, use tr:tab/0:
31> tr:tab().
trace
To switch to a new table, use tr:set_tab/1. The table need not exist.
32> tr:set_tab(tmp).
ok
Now you can collect traces to the new table without changing the original one.
33> tr:trace([lists]), lists:seq(1, 10), tr:stop_tracing().
ok
34> tr:select().
[#tr{index = 1, pid = <0.175.0>, event = call,
     mfa = {lists, ukeysort, 2},
     data = [1,
             [{'Traces', [#tr{index = 2, pid = <0.175.0>, event = call,
                             mfa = {tr_SUITE, sleepy_factorial, 1},
                             data = [2],
(...)
You can dump a table to file with tr:dump/1 - let's dump the tmp table:
35> tr:dump("tmp.ets").
ok
In a new Erlang session we can load the data with tr:load/1. This will set the current table name to tmp.
1> tr:start().
{ok, <0.181.0>}
2> tr:load("tmp.ets").
{ok, tmp}
3> tr:select().
(...)
4> tr:tab().
tmp
Finally, you can remove all traces from the ETS table with tr:clean/0.
5> tr:clean().
ok
To stop tr, just call tr:stop/0.
Example use cases
Debugging a vague error
While reworking the LDAP connection layer in MongooseIM, the following error occured in the logs:
14:46:35.002 [warning] lager_error_logger_h dropped 79 messages in the last second that exceeded the limit of 50 messages/sec
14:46:35.002 [error] gen_server 'wpool_pool-mongoose_wpool$ldap$global$bind-1' terminated with reason: no case clause matching {badkey,handle} in wpool_process:handle_info/2 line 123
14:46:35.003 [error] CRASH REPORT Process 'wpool_pool-mongoose_wpool$ldap$global$bind-1' with 1 neighbours crashed with reason: no case clause matching {badkey,handle} in wpool_process:handle_info/2 line 123
14:46:35.003 [error] Supervisor 'wpool_pool-mongoose_wpool$ldap$global$bind-process-sup' had child 'wpool_pool-mongoose_wpool$ldap$global$bind-1' started with wpool_process:start_link('wpool_pool-mongoose_wpool$ldap$global$bind-1', mongoose_ldap_worker, [{port,3636},{encrypt,tls},{tls_options,[{verify,verify_peer},{cacertfile,"priv/ssl/cacert.pem"},...]}], [{queue_manager,'wpool_pool-mongoose_wpool$ldap$global$bind-queue-manager'},{time_checker,'wpool_pool-mongoose_wpool$ldap$global$bind-time-checker'},...]) at <0.28894.0> exit with reason no case clause matching {badkey,handle} in wpool_process:handle_info/2 line 123 in context child_terminated
14:46:35.009 [info] Connected to LDAP server
14:46:35.009 [error] gen_server 'wpool_pool-mongoose_wpool$ldap$global$default-1' terminated with reason: no case clause matching {badkey,handle} in wpool_process:handle_info/2 line 123
14:46:35.009 [error] CRASH REPORT Process 'wpool_pool-mongoose_wpool$ldap$global$default-1' with 1 neighbours crashed with reason: no case clause matching {badkey,handle} in wpool_process:handle_info/2 line 123
As this messages appear every 10 seconds (on each attempt to reconnect to LDAP), we can start tracing.
The most lkely culprit is the mongoose_ldap_worker module, so let's trace it:
(mongooseim@localhost)16> tr:trace([mongoose_ldap_worker]).
ok
A few seconds (and error messages) later we can check the traces for the badkey value we saw in the logs:
(mongooseim@localhost)17> tr:filter(fun(T) -> tr:contains_data(badkey, T) end).
[#tr{index = 255, pid = <0.8118.1>, event = exception,
     mfa = {mongoose_ldap_worker, connect, 1},
     data = {error, {badkey, handle}},
     ts = 1557838064073778},
     (...)
This means that the key handle was missing from a map.
Let's see the traceback to find the exact place in the code:
(mongooseim@localhost)18> tr:traceback(fun(T) -> tr:contains_data(badkey, T) end).
[#tr{index = 254, pid = <0.8118.1>, event = call,
     mfa = {mongoose_ldap_worker, connect, 1},
     data = [#{connect_interval => 10000, encrypt => tls, password => <<>>,
               port => 3636, root_dn => <<>>,
               servers => ["localhost"],
               tls_options =>
                   [{verify, verify_peer},
                    {cacertfile, "priv/ssl/cacert.pem"},
                    {certfile, "priv/ssl/fake_cert.pem"},
                    {keyfile, "priv/ssl/fake_key.pem"}]}],
     ts = 1557838064052121}, ...]
We can see that the handle key is missing from the map passed to mongoose_ldap_worker:connect/1.
After looking at the source code of this function and searching for handle we can see only one matching line:
                    State#{handle := Handle};
The := operator assumes that the key is already present in the map.
The solution would be to either change it to => or ensure that the map already contains that key.
Loading traces to the trace table after tracing to file
It's possible to use tr with a file generated by dbg:trace_port/2 tracing.
The file may be generated on another system.
1> {ok, St} = tr:init({}).
{ok, #{index => 0, traced_modules => []}}
2> dbg:trace_client(file, "/Users/erszcz/work/myproject/long-pong.dbg.trace", {fun tr:handle_trace/2, St}).
<0.178.0>
3> tr:select().
[#tr{index = 1, pid = <14318.7477.2537>, event = call,
     mfa = {mod_ping, user_ping_response_metric, 3},
     data = [{jid, <<"user1">>, <<"myproject.com">>, <<"res1">>,
              <<"user1">>, <<"myproject.com">>, <<"res1">>},
             {iq, <<"EDC1944CF88F67C6">>, result, <<>>, <<"en">>, []},
             5406109],
     ts = 1553517330696515},
...

Unique, auto-incremented identifier of a tr() record.

          -type init_options() :: #{tab => table(), index => index(), limit => limit()}.


      


Initialization options.
tab is the ETS table used for storing traces (default: trace). index is the index value of the first inserted trace (default: 1). When size of tab reaches the optional limit, tracing is stopped.

Maximum number of items.

Message event types to trace. Default: none.

  



  
    
      
    
    
      mfargs/0



        
          
        

    

  


  

      

          -type mfargs() :: {module(), atom(), list()}.


      


Module, function and arguments.

Specifies traced modules and/or individual functions. Default: [].

Condition checked before collecting message traces for a process.
after_traced_call (default) means that a process needs to call at least one traced function before its message events start being collected. always means that messages for all traced processes are collected.

Total own time (without other called functions).

A list of processes to trace. Default: all.

          -type pred(T) :: fun((T) -> boolean()).


      


Predicate returning true for matching terms of type T.
For other terms, it can return a different value or fail.

          -type prev_next_options() :: #{tab => table(), pred => pred(tr())}.


      


Options for obtaining previous and next traces.
tab is the ETS table with traces (default: trace). pred allows to filter the matching traces (default: allow any trace)

Options for trace ranges.
Optional limit is the maximum depth of calls in the returned ranges. All traces (including messages) exceeding that depth are skipped.

Which ranges to return. Incomplete ranges are missing at least one return. By default, all ranges are returned.

Recipient pid with a boolean indicating if it exists.

          -type result() :: {return | exception, any()}.


      


Result of a function call.

          -type selector(Data) :: fun((tr()) -> Data).


      


Trace selector function.
For selected traces, it returns Data. For other traces, it should fail.

          -type state() ::
          #{tab := table(),
            index := index(),
            limit := limit(),
            trace := none | trace_spec(),
            tracer_pid := none | pid()}.

ETS table name.

          -type tb_acc() :: tb_acc_tree() | tb_acc_list().

          -type tb_acc_list() :: [[tr()]].

          -type tb_acc_tree() :: [{tr(), tb_acc_tree()}].

          -type tb_format() :: list | tree | root.


      


Format in which tracebacks are returned.
list (default) returns a list of tracebacks. tree merges them into a list of trees. root returns only the root of each tree.

          -type tb_options() ::
          #{tab => tr_source(),
            output => tb_output(),
            format => tb_format(),
            order => tb_order(),
            limit => limit()}.


      


Traceback options.
Optional limit is the maximum number of tracebacks to collect before filtering them according to output.

          -type tb_order() :: top_down | bottom_up.


      


Order of calls in each returned traceback. Default: top_down.

          -type tb_output() :: shortest | longest | all.


      


Which tracebacks to return if they overlap. Default: shortest.

          -type tb_tree() :: tr() | {tr(), [tb_tree()]}.


      


Multiple tracebacks with a common root merged into a tree structure.

          -type top_call_trees_options() ::
          #{max_size => pos_integer(),
            min_count => call_tree_count(),
            min_time => acc_time(),
            output => top_call_trees_output()}.


      


Options for repeated call tree statistics.
min_time is an optional minimum accumulated time of a tree. min_count (default: 2) specifies minimum number of repetitions of a tree. max_size (default: 10) specifies maximum number of listed call trees.

          -type top_call_trees_output() :: reduced | complete.


      


Specifies the behaviour for overlapping call trees.
reduced (default) hides subtrees, while complete keeps them.

          -type tr() ::
          #tr{index :: index(),
              pid :: pid(),
              event :: call | return | exception | send | recv,
              mfa :: mfa() | no_mfa,
              data :: term(),
              ts :: integer(),
              info :: recipient() | no_info}.


      


Trace record, storing one collected trace event.
Record fields:
	index - index()
	pid - process in which the traced event occurred, erlang:pid()
	event - call, return or exception for function traces; send or recv for messages.
	mfa - erlang:mfa() for function traces; no_mfa for messages.
	data - Argument list (for calls), returned value (for returns) or class and value (for exceptions).
	ts - Timestamp in microseconds.
	info - For send events it is a recipient() tuple; otherwise no_info.

          -type tr_source() :: table() | [tr()].


      


Source of traces: an ETS table or a list of traces. Default: tab().

  



  
    
      
    
    
      trace_options/0



        
          
        

    

  


  

      

          -type trace_options() ::
          #{modules => module_spec(),
            pids => pids(),
            msg => message_event_types(),
            msg_trigger => msg_trigger()}.


      


Options for tracing.

          -type trace_spec() ::
          #{modules := module_spec(),
            pids := pids(),
            msg := message_event_types(),
            msg_trigger := msg_trigger()}.

          -type traced_pids_tab() :: none | ets:table().

          -type tree() ::
          #node{module :: module(),
                function :: atom(),
                args :: list(),
                children :: [#node{}],
                result :: result()}.


      


Function call tree node.
Record fields:
	module - module name
	function - function name
	args - argument list
	children - a list of child nodes, each of them being tree()
	result - return value or exception, result()

          -type tree_item() :: {acc_time(), call_tree_count(), tree()}.


      


Function call tree with its accumulated time and number of repetitions.

  


        

      

      
        Functions

        


  
    
      
    
    
      app_modules(AppName)



        
          
        

    

  


  

      

          -spec app_modules(atom()) -> [module()].


      


Returns all module names for an application.

          -spec call_stat(selector(Key)) -> #{Key => {call_count(), acc_time(), own_time()}}.


      


Returns call time statistics for traces selected from tab().
See also: call_stat/2.

          -spec call_stat(selector(Key), tr_source()) -> #{Key => {call_count(), acc_time(), own_time()}}.


      


Returns call time statistics for traces selected from tr_source().
Calls are aggregated by Key returned by KeyF.

          -spec clean() -> ok.


      


Removes all traces from the current ETS table.

          -spec contains_data(term(), tr()) -> boolean().


      


Returns traces containing DataVal in #tr.data.
DataVal can occur in (possibly nested) tuples, maps or lists.

          -spec contains_val(T, T) -> boolean().


      


Checks if the given value DataVal is present within Data.
Returns true if DataVal is found, otherwise returns false. DataVal can occur in (possibly nested) tuples, maps or lists.

  



  
    
      
    
    
      do(Index)



        
          
        

    

  


  

      

          -spec do(tr()) -> term().


      


Executes the function call for the provided tr() record or index.

          -spec dump(file:name_all()) -> ok | {error, any()}.


      


Dumps the tab() table to a file.

          -spec filter(pred(tr())) -> [tr()].


      


Returns matching traces from tab().

  



  
    
      
    
    
      filter(F, Tab)



        
          
        

    

  


  

      

          -spec filter(pred(tr()), tr_source()) -> [tr()].


      


Returns matching traces from tr_source().

          -spec load(file:name_all()) -> {ok, table()} | {error, any()}.


      


Loads an ETS trace table from a file, and makes it the current table.
Overwrites the current table if it is empty.

          -spec lookup(index()) -> tr().


      


Returns the tr() record from tab() for an index.

          -spec match_data(pred(term()), tr()) -> boolean().


      


Returns traces with #tr.data containing any values matching the provided predicate.
The matching values can occur in (possibly nested) tuples, maps or lists.

  



  
    
      
    
    
      match_val(Pred, T)



        
          
        

    

  


  

      

          -spec match_val(pred(T), T) -> boolean().


      


Checks if Val contains any values matching the predicate Pred.
The matching values can occur in (possibly nested) tuples, maps or lists.

  



  
    
      
    
    
      mfargs(_, Args)



        
          
        

    

  


  

      

          -spec mfargs(mfa(), list()) -> mfargs().


      


Replaces arity with Args in an MFA tuple.

  



  
    
      
    
    
      mfarity(_)



        
          
        

    

  


  

      

          -spec mfarity(mfa() | mfargs()) -> mfa().


      


Replaces arguments with arity in an MFA tuple.

  



  
    
      
    
    
      next(Index)



        
          
        

    

  


  

      

          -spec next(index() | tr()) -> tr().


      



  



  
    
      
    
    
      next(Index, Options)



        
          
        

    

  


  

      

          -spec next(index() | tr(), prev_next_options()) -> tr().


      


Returns the next trace after the provided index or a trace record. The traces can be filtered by the provided predicate.

  



  
    
      
    
    
      prev(Index)



        
          
        

    

  


  

      

          -spec prev(index() | tr()) -> tr().


      



  



  
    
      
    
    
      prev(Index, Options)



        
          
        

    

  


  

      

          -spec prev(index() | tr(), prev_next_options()) -> tr().


      


Returns the previous trace before the provided index or a trace record. The traces can be filtered by the provided predicate.

  



  
    
      
    
    
      print_sorted_call_stat(KeyF, Limit)



        
          
        

    

  


  

      

          -spec print_sorted_call_stat(selector(_), limit()) -> ok.


      


Prints sorted call time statistics for the selected traces from tab().
The statistics are sorted according to acc_time(), descending. Only top Limit rows are printed.
See also: sorted_call_stat/1.

  



  
    
      
    
    
      range(PredF)



        
          
        

    

  


  

      

          -spec range(pred(tr()) | index() | tr()) -> [tr()].


      


Returns a list of traces from tab() between the first matched call and the corresponding return.
Matching can be done with a predicate function, an index value or a tr() record. Fails if no trace is matched.
See also: range/2.

  



  
    
      
    
    
      range(Index, Options)



        
          
        

    

  


  

      

          -spec range(pred(tr()) | index() | tr(), range_options()) -> [tr()].


      


Returns a list of traces from tr_source() between the first matched call and the corresponding return.
Fails if no call is matched.

  



  
    
      
    
    
      ranges(PredF)



        
          
        

    

  


  

      

          -spec ranges(pred(tr())) -> [[tr()]].


      


Returns lists of traces from tab() between matched calls and corresponding returns.
See also: ranges/2.

  



  
    
      
    
    
      ranges(PredF, Options)



        
          
        

    

  


  

      

          -spec ranges(pred(tr()), range_options()) -> [[tr()]].


      


Returns lists of traces from tr_source() between matched calls and corresponding returns.

  



  
    
      
    
    
      reduce_call_trees(TreeTab)



        
          
        

    

  


  

      

          -spec reduce_call_trees(ets:tid()) -> true.


      



  



  
    
      
    
    
      root(Tr)



        
          
        

    

  


  

      

          -spec root(tb_tree()) -> tr().


      


Returns the root call of the provided tb_tree().

  



  
    
      
    
    
      roots(Trees)



        
          
        

    

  


  

      

          -spec roots([tb_tree()]) -> [tr()].


      


Returns the root call of each tb_tree() from the provided list.

  



  
    
      
    
    
      select()



        
          
        

    

  


  

      

          -spec select() -> [tr()].


      


Returns a list of all collected traces from tab().

  



  
    
      
    
    
      select(F)



        
          
        

    

  


  

      

          -spec select(selector(Data)) -> [Data].


      


Selects data from matching traces from tab() with ets:fun2ms(F).

  



  
    
      
    
    
      select(F, DataVal)



        
          
        

    

  


  

      

          -spec select(selector(Data), term()) -> [Data].


      


Selects data from matching traces from tab() with ets:fun2ms(F).
Additionally, the selected traces have to contain DataVal in #tr.data. DataVal can occur in (possibly nested) tuples, maps or lists.

  



  
    
      
    
    
      seq_next(Index)



        
          
        

    

  


  

      

          -spec seq_next(index() | tr()) -> tr().


      



  



  
    
      
    
    
      seq_prev(Index)



        
          
        

    

  


  

      

          -spec seq_prev(index() | tr()) -> tr().


      



  



  
    
      
    
    
      set_tab(Tab)



        
          
        

    

  


  

      

          -spec set_tab(table()) -> ok.


      


Sets a new ETS table for collecting traces, creating it if it doesn't exist.

  



  
    
      
    
    
      sorted_call_stat(KeyF)



        
          
        

    

  


  

      

          -spec sorted_call_stat(selector(Key)) -> [{Key, call_count(), acc_time(), own_time()}].


      


Returns sorted call time statistics for the selected traces from tab().
The statistics are sorted according to acc_time(), descending.
See also: call_stat/1.

  



  
    
      
    
    
      start()



        
          
        

    

  


  

      

          -spec start() -> {ok, pid()}.


      


Starts tr as a stand-alone gen_server. Intended for interactive use.
See also: start/1.

  



  
    
      
    
    
      start(Opts)



        
          
        

    

  


  

      

          -spec start(init_options()) -> {ok, pid()}.


      


Starts tr as a stand-alone gen_server. Intended for interactive use.
You can override the selected Opts.

  



  
    
      
    
    
      start_link()



        
          
        

    

  


  

      

          -spec start_link() -> {ok, pid()}.


      


Starts tr as part of a supervision tree.
See also: start/1.

  



  
    
      
    
    
      start_link(Opts)



        
          
        

    

  


  

      

          -spec start_link(init_options()) -> {ok, pid()}.


      


Start tr as part of a supervision tree.
See also: start/1.

  



  
    
      
    
    
      stop()



        
          
        

    

  


  

      

          -spec stop() -> ok.


      


Stops the whole tr server process.

  



  
    
      
    
    
      stop_tracing()



        
          
        

    

  


  

      

          -spec stop_tracing() -> ok | {error, not_tracing}.


      


Stops tracing, disabling all trace specs.
Any future messages from the Erlang tracer will be ignored.

  



  
    
      
    
    
      tab()



        
          
        

    

  


  

      

          -spec tab() -> table().


      


Returns the name of the current ETS trace table in use.

  



  
    
      
    
    
      top_call_trees()



        
          
        

    

  


  

      

          -spec top_call_trees() -> [tree_item()].


      


Returns statistics of repeated function call trees that took most time.
See also: top_call_trees/1.

  



  
    
      
    
    
      top_call_trees(Options)



        
          
        

    

  


  

      

          -spec top_call_trees(top_call_trees_options()) -> [tree_item()].


      


Returns statistics of repeated function call trees that took most time.
Two call trees repeat if they contain the same function calls and returns in the same order taking the same arguments and returning the same values, respectively. The results are sorted according to accumulated time.

  



  
    
      
    
    
      trace(Modules)



        
          
        

    

  


  

      

          -spec trace(module_spec() | trace_options()) -> ok | {error, already_tracing}.


      


Starts tracing of the specified functions/modules and/or message events.
You can either provide a list of modules/functions or a more generic map of options.

  



  
    
      
    
    
      trace(Modules, Pids)



        
          
        

    

  


  

      

          -spec trace(module_spec(), pids()) -> ok | {error, already_tracing}.


      


Starts tracing of the specified functions/modules in specific processes.

  



  
    
      
    
    
      trace_app(App)



        
          
        

    

  


  

      

          -spec trace_app(atom()) -> ok | {error, already_tracing}.


      


Starts tracing of all modules in an application.

  



  
    
      
    
    
      trace_apps(Apps)



        
          
        

    

  


  

      

          -spec trace_apps([atom()]) -> ok | {error, already_tracing}.


      


Starts tracing of all modules in all provided applications.

  



  
    
      
    
    
      traceback(Pred)



        
          
        

    

  


  

      

          -spec traceback(pred(tr()) | index() | tr()) -> [tr()].


      


Returns traceback of the first matching trace from tr_source().
Matching can be done with a predicate function, an index value or a tr record. Fails if no trace is matched.
See also: traceback/2.

  



  
    
      
    
    
      traceback(Index, Options)



        
          
        

    

  


  

      

          -spec traceback(pred(tr()) | index() | tr(), tb_options()) -> [tr()].


      


Returns traceback of the first matching trace from tr_source().
Fails if no trace is matched. The options limit and format do not apply.

  



  
    
      
    
    
      tracebacks(PredF)



        
          
        

    

  


  

      

          -spec tracebacks(pred(tr())) -> [[tr()]] | [tb_tree()].


      


Returns tracebacks of all matching traces from tab().
See also: tracebacks/2.

  



  
    
      
    
    
      tracebacks(PredF, Options)



        
          
        

    

  


  

      

          -spec tracebacks(pred(tr()), tb_options()) -> [[tr()]] | [tb_tree()].


      


Returns tracebacks of all matching traces from tr_source().

  



  
    
      
    
    
      ts(Tr)



        
          
        

    

  


  

      

          -spec ts(tr()) -> string().


      


Returns human-readable timestamp according to RFC 3339.

  


        

      


  OEBPS/dist/epub-4WIP524F.js
(()=>{var s=document.querySelector.bind(document),o=document.querySelectorAll.bind(document);function r(e){document.readyState!=="loading"?e():document.addEventListener("DOMContentLoaded",e)}var l="hll";window.addEventListener("exdoc:loaded",t);function t(){o("[data-group-id]").forEach(e=>{e.addEventListener("mouseenter",i),e.addEventListener("mouseleave",i)})}function i(e){let n=e.currentTarget,a=e.type==="mouseenter",c=n.getAttribute("data-group-id");n.parentElement.querySelectorAll(`[data-group-id="${c}"]`).forEach(u=>{u.classList.toggle(l,a)})}r(()=>{t()});})();




